Differentialgleichungssysteme

- ## Hausaufgabe 1: Beispiel 3.9

In[31]:=
```
DSolve[{y1'[x] == 3 y1[x] - x, y2'[x] == y1[x] - y2[x], y1[0] == 2, y2[0] == 1},
  {y1[x], y2[x]}, x]
```

Out[31]= $\left\{\left\{y1(x) \to \frac{1}{9}\left(3x + 17e^{3x} + 1\right),\ y2(x) \to \frac{1}{36}e^{-x}\left(12e^{x}x - 8e^{x} + 17e^{4x} + 27\right)\right\}\right\}$

In[32]:=
```
Simplify[%]
```

Out[32]= $\left\{\left\{y1(x) \to \frac{1}{9}\left(3x + 17e^{3x} + 1\right),\ y2(x) \to \frac{1}{36}\left(12x + 27e^{-x} + 17e^{3x} - 8\right)\right\}\right\}$

- **Schrittweise: Die erste Differentialgleichung löst man mit dem Ansatzverfahren**

In[33]:=
```
sol = DSolve[{y1'[x] == 3 y1[x] - x, y1[0] == 2}, y1[x], x]
```

Out[33]= $\left\{\left\{y1(x) \to \frac{1}{9}\left(3x + 17e^{3x} + 1\right)\right\}\right\}$

- **Einsetzen von y**

In[34]:=
```
y2'[x] == y1[x] - y2[x] /. sol[[1]]
```

Out[34]= $y2'(x) = \frac{1}{9}\left(3x + 17e^{3x} + 1\right) - y2(x)$

- **ergibt wieder eine inhomogene Differentialgleichung, die mit dem Ansatzverfahren gelöst werden kann**

In[35]:=
```
Simplify[DSolve[{y2'[x] == y1[x] - y2[x] /. sol[[1]], y2[0] == 1}, y2[x], x]]
```

Out[35]= $\left\{\left\{y2(x) \to \frac{1}{36}\left(12x + 27e^{-x} + 17e^{3x} - 8\right)\right\}\right\}$

- ## Hausaufgabe 2: Löse y'=A x

In[77]:=
```
A = {{7, -1, 6}, {-10, 4, -12}, {-2, 1, -1}}
```

Out[77]= $\begin{pmatrix} 7 & -1 & 6 \\ -10 & 4 & -12 \\ -2 & 1 & -1 \end{pmatrix}$

In[78]:=
```
CharacteristicPolynomial[A, λ]
```

Out[78]= $-\lambda^3 + 10\lambda^2 - 31\lambda + 30$

In[79]:=
```
Eigenvalues[A]
```

Out[79]= $\{5, 3, 2\}$

In[80]:=
```
system = Eigensystem[A]
```

Out[80]= $\begin{pmatrix} 5 & 3 & 2 \\ \{-3, 6, 2\} & \{-1, 2, 1\} & \{-1, 1, 1\} \end{pmatrix}$

In[81]:=
```
Transpose[system[[2]]]
```

Out[81]= $\begin{pmatrix} -3 & -1 & -1 \\ 6 & 2 & 1 \\ 2 & 1 & 1 \end{pmatrix}$

- **Somit ist die allgemeine Lösung des homogenen Systems gegeben durch**

In[82]:=
```
homogenelösung = MapThread[Rule,
  {{y1[x], y2[x], y3[x]}, Transpose[system[[2]]].({K, L, M} * Map[Exp, system[[1]] * x])}]
```

Out[82]= $\left\{y1(x)\to -3\,K\,e^{5x}-L\,e^{3x}-M\,e^{2x},\ y2(x)\to 6\,K\,e^{5x}+2\,L\,e^{3x}+M\,e^{2x},\ y3(x)\to 2\,K\,e^{5x}+L\,e^{3x}+M\,e^{2x}\right\}$

- **Probe durch Einsetzen:**

In[83]:=
```
{{y1'[x]}, {y2'[x]}, {y3'[x]}} - A.{{y1[x]}, {y2[x]}, {y3[x]}} /.
  Flatten[{homogenelösung, Map[D[#, x] &, homogenelösung]}] // Simplify
```

Out[83]= $\begin{pmatrix} 0 \\ 0 \\ 0 \end{pmatrix}$

In[84]:=
```
mathematicalösung =
 DSolve[{y1'[x] == 7 y1[x] - y2[x] + 6 y3[x], y2'[x] == -10 y1[x] + 4 y2[x] - 12 y3[x],
   y3'[x] == -2 y1[x] + y2[x] - y3[x]}, {y1[x], y2[x], y3[x]}, x]
```

Out[84]= $\{\{y1(x)\to c_1\,e^{2x}\left(-4\,e^x+3\,e^{3x}+2\right)-c_2\,e^{2x}\,(e^x-1)+3\,c_3\,e^{3x}\left(e^{2x}-1\right),$
$y2(x)\to -2\,c_1\,e^{2x}\left(-4\,e^x+3\,e^{3x}+1\right)+c_2\,e^{2x}\,(2\,e^x-1)-6\,c_3\,e^{3x}\left(e^{2x}-1\right),$
$y3(x)\to -2\,c_1\,e^{2x}\left(-2\,e^x+e^{3x}+1\right)+c_2\,e^{2x}\,(e^x-1)-c_3\,e^{3x}\left(2\,e^{2x}-3\right)\}\}$

- **ebenfalls Probe durch Einsetzen:**

In[85]:=
```
{{y1'[x]}, {y2'[x]}, {y3'[x]}} - A.{{y1[x]}, {y2[x]}, {y3[x]}} /.
  Flatten[{mathematicalösung, Map[D[#, x] &, mathematicalösung]}] // Simplify
```

Out[85]= $\begin{pmatrix} 0 \\ 0 \\ 0 \end{pmatrix}$

- ***Mathematica* verwendet die Matrix-Exponentialfunktion**

In[86]:=
```
A
```

Out[86]= $\begin{pmatrix} 7 & -1 & 6 \\ -10 & 4 & -12 \\ -2 & 1 & -1 \end{pmatrix}$

In[87]:=
```
MatrixExp[A x]
```

Out[87]= $\begin{pmatrix} e^{2x}\,(2-4\,e^x+3\,e^{3x}) & -e^{2x}\,(-1+e^x) & 3\,e^{3x}\,(-1+e^{2x}) \\ -2\,e^{2x}\,(1-4\,e^x+3\,e^{3x}) & e^{2x}\,(-1+2\,e^x) & -6\,e^{3x}\,(-1+e^{2x}) \\ -2\,e^{2x}\,(1-2\,e^x+e^{3x}) & e^{2x}\,(-1+e^x) & -e^{3x}\,(-3+2\,e^{2x}) \end{pmatrix}$

- **Wronskimatrix**

In[89]:=
```
W = Transpose[system[[2]] * Exp[system[[1]] * x]]
```

Out[89]= $\begin{pmatrix} -3\,e^{5x} & -e^{3x} & -e^{2x} \\ 6\,e^{5x} & 2\,e^{3x} & e^{2x} \\ 2\,e^{5x} & e^{3x} & e^{2x} \end{pmatrix}$

In[90]:=
```
Det[W]
```

Out[90]= $-e^{10x}$

- **Formeln aus der Vorlesung und inhomogenes Problem: Löse y'=A x+b mit**

In[91]:=
```
b = {{2 - 7 x}, {10 x - 4}, {2 x - 1}}
```

Out[91]= $\begin{pmatrix} 2-7\,x \\ 10\,x-4 \\ 2\,x-1 \end{pmatrix}$

- **Variation der Konstanten**

In[92]:=
```
ansatz = MapThread[Rule, {{y1[x], y2[x], y3[x]},
    Transpose[system[[2]]].({K[x], L[x], M[x]} * Map[Exp, system[[1]] * x])}]
```

Out[92]= $\{y1(x) \to -3 e^{5x} K[x] - e^{3x} L(x) - e^{2x} M(x), y2(x) \to 6 e^{5x} K[x] + 2 e^{3x} L(x) + e^{2x} M(x), y3(x) \to 2 e^{5x} K[x] + e^{3x} L(x) + e^{2x} M(x)\}$

- **Einsetzen in die inhomogene Differentialgleichung liefert:**

In[93]:=
```
dgl = {y1'[x], y2'[x], y3'[x]} - A.{y1[x], y2[x], y3[x]} - {2 - 7 x, 10 x - 4, 2 x - 1} /.
    Flatten[{ansatz, Map[D[#, x] &, ansatz]}] // Simplify
```

Out[93]= $\{-3 e^{5x} K'(x) - e^{3x} L'(x) - e^{2x} M'(x) + 7x - 2, 6 e^{5x} K'(x) + 2 e^{3x} L'(x) + e^{2x} M'(x) - 10x + 4, 2 e^{5x} K'(x) + e^{3x} L'(x) + e^{2x} M'(x) - 2x + 1\}$

In[94]:=
```
DSolve[Map[# == 0 &, dgl], {K[x], L[x], M[x]}, x]
```

Out[94]= $\left\{\left\{K[x] \to c_1 - e^{-5x} x,\ L(x) \to c_2 + e^{-3x}(4x+1),\ M(x) \to c_3 + 4 e^{-2x}\left(-\frac{x}{2} - \frac{1}{4}\right)\right\}\right\}$

- **Lösung mit der Lösungsformel aus der Vorlesung. Das Fundamentalsystem (Lösung des homogenen Systems) ist gegeben durch**

In[95]:=
```
W = Transpose[system[[2]] * Exp[system[[1]] * x]]
```

Out[95]= $\begin{pmatrix} -3 e^{5x} & -e^{3x} & -e^{2x} \\ 6 e^{5x} & 2 e^{3x} & e^{2x} \\ 2 e^{5x} & e^{3x} & e^{2x} \end{pmatrix}$

In[96]:=
```
Inverse[W]
```

Out[96]= $\begin{pmatrix} -e^{-5x} & 0 & -e^{-5x} \\ 4 e^{-3x} & e^{-3x} & 3 e^{-3x} \\ -2 e^{-2x} & -e^{-2x} & 0 \end{pmatrix}$

In[97]:=
```
Inverse[W /. x → 0]
```

Out[97]= $\begin{pmatrix} -1 & 0 & -1 \\ 4 & 1 & 3 \\ -2 & -1 & 0 \end{pmatrix}$

In[98]:=
```
Inverse[W].b // Expand
```

Out[98]= $\begin{pmatrix} 5 e^{-5x} x - e^{-5x} \\ e^{-3x} - 12 e^{-3x} x \\ 4 e^{-2x} x \end{pmatrix}$

In[99]:=
```
lösung1 =
  W.(Inverse[W /. x → 0].{{y10}, {y20}, {y30}} + ∫_0^x (Inverse[W].b /. x → t) dt) // Simplify
```

Out[99]= $\begin{pmatrix} x + e^{2x}(2\,y10 + y20 - 1) + 3 e^{5x}(y10 + y30) - e^{3x}(4\,y10 + y20 + 3\,y30 - 1) \\ -e^{2x}(2\,y10 + y20 - 1) - 6 e^{5x}(y10 + y30) + 2 e^{3x}(4\,y10 + y20 + 3\,y30 - 1) + 1 \\ -e^{2x}(2\,y10 + y20 + 2 e^{3x}(y10 + y30) - e^{x}(4\,y10 + y20 + 3\,y30 - 1) - 1) \end{pmatrix}$

- **Vergleich mit DSolve:**

In[100]:=
```
lösung2 = DSolve[{y1'[x] == 7 y1[x] - y2[x] + 6 y3[x] + 2 - 7 x, y2'[x] ==
      -10 y1[x] + 4 y2[x] - 12 y3[x] + 10 x - 4, y3'[x] == -2 y1[x] + y2[x] - y3[x] + 2 x - 1,
    y1[0] == y10, y2[0] == y20, y3[0] == y30}, {y1[x], y2[x], y3[x]}, x] // Simplify
```

Out[100]= $\{\{y1(x) \to -e^{3x}(4\,y10 + y20 + 3\,y30 - 1) + e^{2x}(2\,y10 + y20 - 1) + 3 e^{5x}(y10 + y30) + x, y2(x) \to 2 e^{3x}(4\,y10 + y20 + 3\,y30 - 1) - e^{2x}(2\,y10 + y20 - 1) - 6 e^{5x}(y10 + y30) + 1, y3(x) \to -e^{2x}(-e^{x}(4\,y10 + y20 + 3\,y30 - 1) + 2 e^{3x}(y10 + y30) + 2\,y10 + y20 - 1)\}\}$

- **Einsetzen in die inhomogene Differentialgleichung liefert:**

In[101]:=
```
{y1'[x], y2'[x], y3'[x]} - A.{y1[x], y2[x], y3[x]} - {2 - 7 x, 10 x - 4, 2 x - 1} /.
  Flatten[{lösung2, Map[D[#, x] &, lösung2]}] // Simplify
```

Out[101]= {0, 0, 0}

- **Dasselbe gilt für unsere gefundene Lösung:**

In[102]:=
```
lösung1 - ({{y1[x]}, {y2[x]}, {y3[x]}} /. lösung2[[1]]) // Simplify
```

Out[102]= $\begin{pmatrix} 0 \\ 0 \\ 0 \end{pmatrix}$

In[103]:=
```
lösung1 = W.(Inverse[W /. x → 0].{{1}, {0}, {2}} + ∫_0^x (Inverse[W].b /. x → t) ⅆt) // Simplify
```

Out[103]= $\begin{pmatrix} x + e^{2x} - 9e^{3x} + 9e^{5x} \\ 1 - e^{2x} + 18e^{3x} - 18e^{5x} \\ e^{2x}(-1 + 9e^{x} - 6e^{3x}) \end{pmatrix}$